	Où placer un script ? 
	

	

	



	 

Dans cette ressource, nous n'allons pas vous apprendre à programmer en objet ou en mode séquentiel, nous allons simplement vous indiquer les deux procédures pour démarrer la programmation d'une animation. Si vous n'avez jamais programmé et/ou n'avez pas un gros besoin d'interactivité dans vos animations, favorisez le mode de programmation séquentielle. Quoi qu'il en soit, nous allons principalement vous proposer des scripts de programmation séquentielle dans les différentes ressources de Yazo.net.

 Mode de programmation séquentielle

 

1. Commencez par cliquez sur une image clé ou sur les côtés de la scène.

2. Cliquez dans la fenêtre Actions.

3. Rédigez votre script. 

Voilà, il ne vous reste plus qu'à saisir votre premier script, mais pour commencer, aidez vous de ce tableau pour savoir ce que doit contenir votre script. Nous distinguons les actions qui sont exécutées en continue et celles qui sont exécutées une seule fois, par ailleurs, nous faisons également la différence entre les actions déclenchées par l'utilisateur (actions manuelles) et celles qui se font automatiquement par le programme.

 

 

Actions Automatiques 

 Actions Manuelles

Une
seule fois 

Script placé en dehors des écouteurs gérant les évènement MouseEvent et KeyboardEvent. Saisissez simplement votre ligne d'instruction sans gestionnaire d'évènement.

Exemple : La première ligne de l'exemple 1 ci-dessous : stop() 

Script placé dans un écouteur gérant les évènement MouseEvent et/ou KeyboardEvent.


Exemple : Voir l'exemple 1 ci-dessous. 

En continue

Script placé dans un écouteur gérant l'évènement Event.ENTER_FRAME. Il est également possible d'utiliser la classe Timer().

Exemple : Voir l'exemple 2 ci-dessous. 

 

 

Exemple 1 : Réalisez une interpolation de mouvement de l'image clé 1 à l'image clé 20 (faire défiler une occurrence de la gauche vers la droite de la scène). Créez un calque supplémentaire sur lequel vous placez un symbole de type Clip. Nommez l'occurrence obtenue, ex. : boutonLancement. Placez le script ci-dessous sur l'une des deux images clé de la première image.

stop();

boutonLancement.addEventListener(MouseEvent.MOUSE_DOWN,relancerTeteLecture);

function relancerTeteLecture(evt:MouseEvent) {

   play();

}

 

Exemple 2 : Dans un nouveau document, placez un symbole sur la scène qui représente un rectangle. Nommez l'occurrence obtenue, ex. : planche.

planche.addEventListener(Event.ENTER_FRAME,tournerPlanche);

function tournerPlanche(evt:Event) {

   planche.rotation = planche.rotation+3

}


Consultez à présent les ressources qui traient des Ecouteurs, des Propriétés et de l'Incrémentation de la valeur d'une propriété afin de pouvoir faire vos premiers essais de programmation en AS3. 

Mode de programmation orientée objet


Vous connaissez déjà la programmation orientée objet et vous aimeriez developper en AS3.

1. Créez une nouvelle animation que vous enregistrez au format .fla.

2. Dans la palette Propriétés, spécifiez le nom de la classe du document. Ex. Main. 

3. Créez un document ActionScript que vous nommez Main.as.

4. Lorsque vous exécutez le raccoruci clavier CTRL-Entrée (PC) ou Commande-Entrée (Mac), vous obtiendrez une prévisualisation, aussi bien à partir du fichier .as que .fla.

Remarque : Si vous n'avez pas besoin de faire appel à l'IDE de Flash car vous développez réellement l'intégralité de votre animation à base de code, utilisez le plug-in Flex pour Eclipse ou faites appel à des solutions logicielles telles que FlashDevelopp ou SciTE. 

 


	Les écouteurs 
	
	
	


	 

Lorsque vous allez ajouter de l'interactivité à vos animations, vous ferez appel à des gestionnaires d'évènements appelés également des écouteurs.

Cette notion de gestionnaire d'évènement est capitale en programmation car c'est elle qui fait la relation entre un évènement (ex. : clic sur une occurrence, la fin du chargement d'une image, etc.) et l'exécution d'une ligne d'instruction.

Un écouteur (appelé également gestionnaire d'évènement) est composé de deux parties :

· La ligne d'instruction qui fait référence à l'instance concernée par l'écouteur et lui assigne un évènement.

· La fonction callBack, c'est à dire la fonction qui est appelée lorsque le déclenchement de l'évènement est constaté par le programme.


Voici un premier exemple qui peut vous paraître abstrait si vous n'avez jamais programmé, mais rassurez-vous, nous cherchons simplement à vous transmettre une terminologie. Les nombreux exemples dans les articles de Yazo.net font si souvent appel à des gestionnaires d'évènements que vous finirez par apprendre à les manipuler.

 

    volet.addEventListener(MouseEvent.MOUSE_DOWN,agrandir);

    function agrandir(evt:MouseEvent) {

       volet.scaleX = volet.scaleX+1;

       volet.scaleY = volet.scaleY+1;

    }


Explications

   1. Le premier mot correspond au nom d'une occurrence (volet) sur laquelle vous souhaitez cliquer (MOUSE_DOWN).
   2. Le mot addEventListener(), qualifié de méthode, sert à faire la relation entre le nom d'occurrence (volet) et l'évènement (MouseEvent.MOUSE_DOWN) que l'écouteur va surveiller.
   3. MouseEvent est l'évènement qui va être surveillé à partir du moment où cette ligne d'instruction va être intégralement exécutée.
   4. Le mot MOUSE_DOWN, qualifié de constante,  sert à préciser le type d'évènement.
   5. Pour terminer, le mot agrandir correspond à la fonction qui va être appelée lorsque l'évènement va se déclencher.


La fonction callBack sera toujours composée de la structure suivante :

    function () {

    }


accompagnée du nom de la fonction…

    function agrandir() {

    }


ainsi que du nom d'un objet (comparable à une variable qui stocke plusieurs informations) qui est typé. En clair, vous devez écrire un mot qui ne contient ni espace, ni caractères accentués ou spéciaux et ne commence pas par une majuscule entre les parenthèses de la fonction. Il doit toujours être suivi de deux points et du même "mot" que vous avez utilisé pour définir l'évènement (dans notre exemple, MouseEvent).

    function agrandir(evt:MouseEvent) {

    }

Note : Dans l'exemple ci-dessus, le mot MouseEvent correspond au type de l'objet qui est défini et contenu entre les parenthèses de la fonction agrandir. Très souvent, sur Internet, vous pourrez lire le mot event au lieu du mot que nous utilisons dans cet exmple et utiliserons dans les exemples à venir, à savoir, evt. Evitez d'utiliser event afin qu'il n'y ait aucune confusion avec le mot Event qui a un tout autre sens.


La ligne d'instruction ci-dessous est qualifiée d'écouteur, on peut alors dire : "enregistrer un écouteur", "définir un écouteur", ou encore "créer un écouteur".

    volet.addEventListener(MouseEvent.MOUSE_DOWN,agrandir); 


 
	Interactivité à la souris 
	
	
	


	 

Abordons ensemble un premier exemple, celui d'une carte à jouer sur laquelle nous allons cliquer pour pouvoir la faire disparaître.

1.    Créez une nouvelle animation que vous enregistrez.

2.    Créez un symbole de type Clip.

3.    Gardez sélectionnée sur la scène, l'occurrence du clip que vous venez de créer et nommez la. Exemple : carte

4.    Cliquez ensuite sur la scène (pas sur l'occurrence) ou sur l'image clé 1 de la fenêtre Scénario.

5.    Cliquez dans la fenêtre Actions.

6.    Saisissez le script ci-dessous.


 

   function masquer(evt:MouseEvent) {

      carte.visible=false;

   }

   carte.addEventListener(MouseEvent.MOUSE_DOWN,masquer)



Dans l'exemple suivant (ci-dessous), deux occurrences ont été placées sur la scène et nommées bouton1 et carte. En cliquant sur bouton1, nous souhaitons masquer l'occurrence intitulée carte, voici le script :

(Cliquez préalablement sur l'image-clé de la timeline puis dans la fenêtre Actions avant de modifier script actuel ou de resaisir celui ci-dessous)

   function masquer(evt:MouseEvent) {

       carte.visible=false;

   }

   bouton1.addEventListener(MouseEvent.MOUSE_DOWN,masquer)


Dans les deux exemples que nous venons d'aborder, nous avons fait apel à la constante MOUSE_DOWN, mais nous aurions pu utiliser d'autres évènements :

·     MOUSE_DOWN

·     MOUSE_UP

·     MOUSE_OVER

·     MOUSE_OUT

·     MOUSE_MOVE

·     DOUBLE_CLICK


N.B. : Pour utiliser l'événement DOUBLE_CLICK, vous devez également régler la propriété doubleClickEnabled  de l'occurrence concernée sur true.

   carre.addEventListener(MouseEvent.DOUBLE_CLICK,tourner);

   carre.doubleClickEnabled = true;

   function tourner(evt:MouseEvent) {

       carre.rotation = 45;

   }

 

La propriété currentTarget


Afin qu'une fonction puisse "servir" à plusieurs occurrences, vous pouvez "cibler" une occurrence avec la propriété currentTarget.

1.    Créez une nouvelle animation que vous enregistrez.

2.    Créez un symbole de type Clip.

3.    Dupliquez deux fois  l'occurrence du clip que vous venez de créer.

4.    Nommez les trois occurrences. Exemple : carre1, carre2, carre3

5.    Cliquez ensuite sur la scène (pas sur l'occurrence) ou sur l'image clé 1 de la fenêtre Scénario.

6.    Saisissez le script ci-dessous.

 

   carre1.addEventListener(MouseEvent.CLICK,tourner);

   carre2.addEventListener(MouseEvent.CLICK,tourner);

   carre3.addEventListener(MouseEvent.CLICK,tourner);

   function tourner(evt:MouseEvent) {

       evt.currentTarget.rotation = 45;

   }


En ayant remplacé carre de l'exemple précédent par evt.currentTarget, nous pouvons utiliser la même fonction avec différentes occurrences.
Quelle est la différence entre currentTarget et target ?

La réponse à cette question va vous sembler évidente dès lors que vous aurez pris connaissance de l'exemple suivant :

Vous venez de placer sur la scène, 3 occurrences intitulées carte1, carte2 et carte3. Vous les sélectionnez toutes les 3 et appuyez sur la touche de fonction F8 pour créer un nouveau symbole. Vous avez à présente sur la scène, non plus trois occurrences, mais une seule, celle de votre nouveau clip. Nommez la ensemble. Afin de bien comprendre la différence entre currentTarget et target, cliquez sur l'image clé de la timeline ou sur la scène (en dehors de l'occurrence, puis cliquez dans la fenêtre Actions. Saisissez à présent ce script :

   ensemble.addEventListener(MouseEvent.MOUSE_DOWN,deplacer);

   ensemble.addEventListener(MouseEvent.MOUSE_UP,relacher);

   function deplacer(evt:MouseEvent) {

       evt.currentTarget.startDrag();

   }


   function relacher(evt:MouseEvent) {

       stopDrag();

   }


Modifiez à présent simplement le paramètre currentTarget par target.

   ensemble.addEventListener(MouseEvent.MOUSE_DOWN,deplacer);

   ensemble.addEventListener(MouseEvent.MOUSE_UP,relacher);

   function deplacer(evt:MouseEvent) {

        evt.target.startDrag();

   }

   function relacher(evt:MouseEvent) {

       stopDrag();

   }


Qu'observez-vous ? Lorsque vous utilisez la propriété currentTarget, cette dernière fait référence à l'occurrence sur laquelle vous avez posé u écouteur, ne tenant pas compte des éventuelles occurrence contenues dans l'occurrence ensemble. Avec la propriété target, le programme traverse l'occurrence ensemble et va réellement cibler l'occurrence cliquée et contenue dans l'occurrence ensemble. Conclusion, si vous souhaitez programmer l'interactivité des occurrence contenues dans une autre occurrence, vous pouvez utiliser target alors qu'avec la propriété currentTarget, vous ciblez celle que vous avez désigné dans la ligne d'instruction ci-dessous.

   ensemble.addEventListener(MouseEvent.MOUSE_DOWN,deplacer);  


	Interactivité au clavier 
	
	
	


	 

Pour pouvoir utiliser le clavier dans une animation (ex. : déplacer un personnage, contrôler des médias, etc.) voici un article qui vous présente des scripts qui font appel à l'évènement KeyboardEvent. Vous découvrirez avec surprise que le gestionnaire d'évènement est très proche de celui qui gère l'interactivité à l'aide de la souris.



Utiliser l'événement KeyboardEvent.KEY_DOWN et la propriété stage au lieu du nom d'une instance.

   stage.addEventListener(KeyboardEvent.KEY_DOWN,avancer);

   function avancer(evt: KeyboardEvent) {

       carre.rotation = carre.rotation+3;

   }


Quelle que soit la touche enfoncée, la ligne d'instruction contenue dans la fonction avancer s'exécute. Pour choisir une touche précise, il faut tester le code de cette dernière.

   stage.addEventListener(KeyboardEvent.KEY_DOWN,avancer);

   function avancer(evt:KeyboardEvent) {

       if(evt.keyCode==37) carre.rotation = carre.rotation-3;

       if(evt.keyCode==39) carre.rotation = carre.rotation+3;

   }

Le script ci-dessus permet de faire tourner une occurrence dans les deux sens d'une aiguille d'une montre. 
Code exécuté en continu 

 

Dans certains cas, vous aurez besoin d'exécuter une ou plusieurs  lignes d'instructions de façon répétée. Exemples : Pour le déplacement d'une occurrence sur la scène, l'aiguille d'une horloge qui tourne, etc. Utilisez l'événement Event.ENTER_FRAME.

   carre.addEventListener(Event.ENTER_FRAME, tourner);

   function tourner(evt:Event) {

       carre.rotation = carre.rotation+3;

   }


Dans cet exemple, nous avons disposé un symbole sur la scène, l'occurrence obtenue a été nommée "carre". 

Devant la fonction addEventListener(), nous avons ciblé l'occurrence intitulée "carte", mais nous aurions pu faire référence à n'importe quelle autre instance, y compris la propriété stage (de l'animtion).

   stage.addEventListener(Event.ENTER_FRAME, tourner);

 


	Propriétés d'une occurrence 
	
	
	


	 

Dans l'article précédent, nous avons incrémenté la rotation d'une occurrence de 3° en 3°, de la même façon, nous aurions pu régler d'autres "paramètres" que nous devons qualifier de propriétés.

Nom

Propriété 

 

Unité

Valeurs mini/maxi 

 

 

 

 

 

 width

 Largeur

 

 pixel

 Aucune

 height

 Hauteur

 

 pixel

 Aucune

 x

 Position hori.

 

 pixel

 Aucune

 y

 Position verti.

 

 pixel

 Aucune

 scaleX

 Echelle hori.

 

 Nombre

Entre - l'infini et + l'infini (Utiliser le chiffre 1 pour une échelle à 100%) 

 scaleY

 Echelle verti.

 

 Nombre

 Idem

 rotation

 Rotation

 

 °

 Aucune 

 alpha

 Transparence

 

 Nombre

 Entre 0 et 1 : Ex. Utiliser 0.5 pour régler une occurrence à 50%.

 visible

 Visibilité

 

 Booléen

 true ou false

 

Observations
· Les propriétés scaleX et scaleY peuvent prendre des valeurs négatives pour effectuer une symétrie (Exemple : carre1.scaleX=-1).

· La propriété rotation peut être réglée à n'importe quelle valeur. En revanche, la lecture de la valeur de cette propriété renvoie -180 au minimum et 180 au maximum.

· Les propriétés x et y peuvent être réglées avec des valeurs dépassant les limites de la scène (valeurs négatives et valeurs plus grandes que la largeur et la hauteur de la scène) afin de masque une occurrence.


Exemple : La fonction trace() ci-dessous renvoie la valeur -170.

 

   carre1.rotation=190

   trace(carre1.rotation)

 

Quelques exemples de lignes d'instructions :

   carre1.rotation=45;

   carre1.scaleX=0.3;

   carre1.visible=false;

   carre1.alpha=0.5;

   carre1.height=50;

   carre1.x=150;

Remarque : Chaque ligne d'instruction ci-dessus doit être testée séparément.

 

Incrémenter et décrémenter la valeur d'une propriété


Il existe deux syntaxes pour augmenter ou diminuer  la valeur d'une propriété ou d'une variable, tout dépend  de la valeur.

Pour une valeur égale à 1 ou -1

   carre.x=carre.x+1

ou 

   carre.x++

et

   carre.x=carre.x-1

ou

   carre.x--



Pour une valeur différente de 1 ou -1

   carre.x = carre.x+3

ou

   carre.x+=3



Des valeurs différentes de 1 ou -1 sous-entendent aussi des valeurs telles que 1,1 ou 0,99
carre.x+=0.8


Exemple :

   carre.addEventListener(MouseEvent.CLICK,avancer);

   function avancer(evt:Event) {

         carre.x = carre.x+3;

   }


